

Information Transfer Between The HP-41 And Series 80 Personal Computers



Summary

The HP 82938A HP-IL Interface for Series 80 personal computers can be used to transfer Series 40 programs plus numeric and ASCII data between Series 40 advanced calculators and Series 80 personal computers. Example programs are included to demonstrate procedures for transferring information.

Equipment Required

Series 40 Advanced Programmable Calculator - 63 registers required

Series 80 personal computer - 32K required

Note: Depending on the type of transfer you are performing when using an HP-86 or HP-87, either a mass storage device or a printer is required.

HP 82160A HP-IL Interface Module (for Series 40)

HP 82938A HP-IL Interface (for Series 80)

Either 00085-15003 Input/Output ROM for the HP-85,

Or 00087-15003 Input/Output ROM for the HP-86 or HP-87

Additional equipment may be required for specific example programs and will be noted at the beginning of each program listing.

Configuration Notes

In all of the examples, the HP-41 acts as the controller and the Series 40 HP-IL module switch must be set to enable. If there is more than one HP-IL peripheral on the loop, the programs assume that the HP-IL Series 80 Interface is the first device.

The ABORTIO 9 statement is used in the Series 80 program to disable the Series 80 HP-IL Interface as system controller. This method of establishing non-controller status is an unsupported feature of the ABORTIO command. You can also set the Series 80 HP-IL Interface to non-controller mode by resetting the system controller switches. To do this, refer to the instructions on pages 9 through 12 in the Owner's Manual for the HP 82938A HP-IL Interface. The ABORTIO 9 statements can then be deleted from the program.

Operation

All of the transfer operations that are printed on the following pages can be executed only when the Series 80 program "4185XFER" has been loaded and is running. This program sets the softkey options on the Series 80 computer screen.

HP Computer Museum
www.hpmuseum.net

For research and education purposes only.

Series 80 BASIC Transfer Program

This program allows transfer of information between the HP-41 and Series 80 personal computers.

This program is written so that each subroutine can be used as a separate program with minor modifications. Dimension statements need to be added to the subroutines if they are to be used alone. It also will be necessary to alter the statement numbers for the error messages in that instance.

4185XFER

	PROGRAM LISTING	COMMENTS
10	DIM B\$(3500),D\$(255),E\$(4200), G\$(4200),H\$(4500)	Variables hold data.
20	ABORTIO 9	Disables the HP 82938A Series 80 Interface as controller.
30	CLEAR	Clears the screen.
40	ON KEY# 1, "41D285" GOTO 140	Labels softkey #1 with 41D285 subroutine.
50	ON KEY# 8, "41DF85" GOTO 370	Labels softkey #8 with 41DF85 subroutine.
60	ON KEY# 2, "41A285" GOTO 490	Labels softkey #2 with 41A285 subroutine.
70	ON KEY# 9, "41AF85" GOTO 690	Labels softkey #9 with 41AF85 subroutine.
80	ON KEY# 3, "41P285" GOTO 890	Labels softkey #3 with 41P285 subroutine.
90	ON KEY# 10, "41PF85" GOTO 1060	Labels softkey #10 with 41PF85 subroutine.
100	ON KEY# 4, "41PRP" GOTO 1180	Labels softkey #4 with 41PRP subroutine.
110	ON KEY# 7, "EXIT" GOTO 1370	Labels softkey #7 with EXIT subroutine.
120	KEY LABEL	
130	GOTO 130	Wait for softkey to be pressed.

Subroutine 41D285

This subroutine stores data from an HP-41 on a Series 80 computer mass storage device.

140	ABORTIO 9	Disables Series 80 HP 82938A HP-IL Interface as controller.
150	CLEAR	Clears the screen.
160	DISP "WAITING FOR DATA..."	
170	ENTER 9; C	Enters number of registers in file.
180	B\$=VAL\$(C)&CHR\$(10)	Stores number of registers in file.
190	CLEAR	Clears the screen.
200	DISP "RECEIVING DATA..."	
210	FOR I=1 TO C	Sets up loop counter to enter data.

PROGRAM LISTING	COMMENTS
220 ENTER 9 ; A\$	Enters each register value from the HP-41.
230 B\$=B\$&A\$&CHR\$(10)	Adds a line feed character to each value and concatenates all values.
240 NEXT I	Goes through loop again.
250 BEEP 100,100	Audible tone to signal data entered.
260 DISP "FILE NAME";	Prompts for file name.
270 INPUT N\$	
280 G=LEN(B\$)+3	
290 ON ERROR GOTO 1310	Checks for duplicate file name error.
300 CREATE N\$,CEIL((G/256*3+G)/256)	Creates file large enough to hold all values.
310 OFF ERROR	
320 ASSIGN# 1 TO N\$	Opens file.
330 PRINT# 1 ; B\$	Writes all register values to mass storage.
340 ASSIGN# 1 TO *	Closes file.
350 DISP "DATA STORED"	
360 GOTO 40	Return to softkey options.

Subroutine 41DF85

This subroutine transfers data stored on a mass storage device to an HP-41.

370 ABORTIO 9	Disables the HP Series 80 HP 82938A HP-IL Interface as controller.
380 CLEAR	Clears screen.
390 B\$=""	Sets B\$ string to null.
400 DISP "FILE NAME";	
410 INPUT N\$	
420 ASSIGN# 1 TO N\$	Opens file.
430 READ# 1 ; B\$	Reads number of registers to be sent and all register values.
440 ASSIGN# 1 TO *	Closes file.
450 DISP "SENDING DATA..."	
460 OUTPUT 9 USING "#,K" ; B\$	Sends values to the HP-41.
470 DISP "DATA SENT"	
480 GOTO 40	Return to softkey options.

Subroutine 41A285

This subroutine transfers ASCII data stored in X-memory to a Series 80 computer.

490 CLEAR	Clears screen.
500 ABORTIO 9	Disables the Series 80 HP 82938A HP-IL Interface as controller.
510 D\$,E\$=""	Sets strings D\$ and E\$ to null.
520 DISP "WAITING FOR DATA..."	
530 ENTER 9 ; D\$	Enters a record of data.
540 IF E\$="" THEN DISP "RECEIVING DATA"	Only display message the first time through the loop.

PROGRAM LISTING	COMMENTS
550 E\$=E&D&CHR\$(10)	Concatenates records and appends a line feed onto each record.
560 IF POS(E\$,CHR\$(254))=0 THEN 530	Character 254 signals the end of the file.
570 BEEP 100,100	Audible tone to signal data entered.
580 DISP "FILE NAME";	Prompt for file name.
590 INPUT N\$	
600 G=LEN(E\$)+3	
610 ON ERROR GOTO 1330	Checks for duplicate file name error.
620 CREATE N\$,CEIL((G/256*3+G)/256)	Creates file large enough to store the ASCII file.
630 OFF ERROR	
640 ASSIGN# 1 TO N\$	Opens file.
650 PRINT# 1; E\$	Writes ASCII file to a storage medium.
660 ASSIGN# 1 TO *	Closes file.
670 DISP "ASCII FILE STORED"	
680 GOTO 40	Return to softkey options.

Subroutine 41AF85

This subroutine transfers ASCII data stored in a file to the HP-41.

690 ABORTIO 9	Disables the Series 80 HP 82938A HP-IL Interface as controller.
700 CLEAR	Clears screen.
710 E\$,G\$=""	Sets strings E\$ and G\$ to null.
720 DISP "FILE NAME";	
730 INPUT N\$	
740 ASSIGN# 1 TO N\$	Opens file.
750 READ# 1; E\$	Reads entire ASCII file into E\$.
760 ASSIGN# 1 TO *	Closes file.
770 R=CEIL(LEN(E\$)/7)	Calculates the number of registers needed for the ASCII file in X-memory.
780 DISP "SENDING DATA..."	
790 OUTPUT 9;R	Sends number of registers needed for the ASCII file to the HP-41.
800 G\$=E\$[1,POS(E\$,CHR\$(10))]	Defines G\$ to be the first record of data.
810 C=CEIL(LEN(G\$)/24)	Calculates the number of alpha registers to be stored for that record.
820 OUTPUT 9;C	Sends the number of alpha registers to be stored for that record.
830 OUTPUT 9 USING "#,K"; G\$	Sends a record of ASCII data.
840 E\$=E\$[POS(E\$,CHR\$(10))+1]	Defines E\$ as E\$ minus the first record.
850 IF NUM(E\$)<>254 THEN 800	Tests to see if all records have been sent.
860 OUTPUT 9;0	If yes, sends 0 to 41 to indicate so.
870 DISP "ASCII FILE SENT"	
880 GOTO 40	Return to softkey options.

Subroutine 41P285

This subroutine receives HP-41 programs and stores them on a mass storage medium.

	PROGRAM LISTING	COMMENTS
890	ABORTIO 9	Disables the Series 80 HP 82938A HP-IL Interface as controller.
900	CLEAR	Clears screen.
910	H\$=""	Sets string H\$ to null.
920	DISP "WAITING FOR PROGRAM..."	
930	ENTER 9 ; H\$	Stores all program data byte values into H\$.
940	BEEP 100,100	Audible tone to signal the data is entered.
950	DISP "FILE NAME";	
960	INPUT N\$	
970	G=LEN (H\$)+3	
980	ON ERROR GOTO 1350	Checks for duplicate file name error.
990	CREATE N\$,CEIL ((G/256*3+G)/256)	Creates a file large enough to hold the entire program.
1000	OFF ERROR	
1010	ASSIGN# 1 TO N\$	Opens file.
1020	PRINT# 1 ; H\$	Writes program data byte values to file.
1030	ASSIGN# 1 TO *	Closes file.
1040	DISP "PROGRAM STORED"	
1050	GOTO 40	Return to softkey options.

Subroutine 41PF85

This subroutine transfers an HP-41 program stored on a mass storage medium to the HP-41.

1060	ABORTIO 9	Disables the Series 80 HP 82938A HP-IL Interface as controller.
1070	CLEAR	Clears the screen.
1080	H\$=""	Sets string H\$ to null.
1090	DISP "FILE NAME";	
1100	INPUT N\$	
1110	ASSIGN# 1 TO N\$	Opens file.
1120	READ# 1 ; H\$	Reads program data byte values into H\$.
1130	ASSIGN# 1 TO *	Closes file.
1140	DISP "SENDING PROGRAM..."	
1150	OUTPUT 9 USING "#,K" ; H\$	Sends program data byte values to the HP-41.
1160	DISP "PROGRAM SENT"	
1170	GOTO 40	Return to softkey options.



Subroutine 41PRP

This subroutine prints HP-41 programs on a printer interfaced to a Series 80 computer.

	PROGRAM LISTING	COMMENTS
1180	CLEAR	Clears the screen.
1190	ABORTIO 9	Disables the Series 80 HP 82938A HP-IL Interface as controller.
1200	A\$,D\$=""	Sets string A\$ and D\$ to null.
1210	DISP "WAITING..."	
1220	ENTER 9 ; D\$	Strips off initial null string.
1230	ENTER 9 ; A\$	Receives data from loop.
1240	A\$=A\$[2]	Strips off carriage return.
1250	PRINT A\$	Print a line of program.
1260	IF POS (A\$, "END") THEN 1280	Checks for end of program.
1270	GOTO 1230	If not the end, get more data.
1280	ENTER 9 USING "#,#B" ; A	Enters final character sent.
1290	DISP "DONE"	Display done if end encountered.
1300	GOTO 40	Return to softkey options.
1310	DISP "DUPLICATE FILE NAME. PICK A NEW ONE."	Error message.
1320	GOTO 260	Get new file name.
1330	DISP "DUPLICATE FILE NAME. PICK A NEW ONE."	Error message.
1340	GOTO 580	Get new file name.
1350	DISP "DUPLICATE FILE NAME. PICK A NEW ONE."	Error message.
1360	GOTO 950	Get new file name.
1370	CLEAR @ END	Clears screen. End of program.

Transferring Data In HP-41 Storage Registers To A Series 80 Computer

This program will transfer data from a Series 40 advanced calculator to a Series 80 mass storage device.

Instructions

1. The data to be transferred must be in the storage registers of the HP-41 main memory.
2. Enter the following program (41D285) into the HP-41. This program requires a minimum SIZE of 001, with the actual SIZE being determined by the amount of data to be transferred.
3. Press key #1 on the Series 80 computer to initiate the subroutine 41D285.
4. To run the program on the HP-41 type in the following:
[XEQ][ALPHA]41D285[ALPHA]

"NO. REG.?" will appear in the display.

Key in the number of registers to be transferred beginning with register 01, then press [R/S].

The display will show "SENDING DATA" and then "DATA SENT" when the transfer is complete.

5. The Series 80 computer will show "RECEIVING DATA" until the transfer is complete. The computer will then beep and ask for the file name under which you want to store the data. After you type in the file name press [END LINE] and "DATA STORED" will appear in the display.

The data will be stored in string B\$ with each piece of data separated by a line feed character. The first value in this string is the number of pieces of data that were transferred.

NOTE:

The maximum number of registers that can be transferred is 306. The time required to transfer 306 registers of data is approximately two minutes.

41D285

PROGRAM LISTING

```
01 LBL "41D285"  
02 CF 17  
  
03 CF 21  
04 CF 29  
05 "NO. REGS?"  
06 PROMPT  
07 CLA  
08 ARCL X  
09 OUTA  
10 1 E3  
11 /  
12 1  
13 +  
14 STO 00  
15 "SENDING DATA"  
16 AVIEW  
17 LBL 00  
18 CLA  
19 ARCL IND 00
```

COMMENTS

Enables carriage return/line feed with OUTA.
Clears print enable flag.
Removes comma as separator.

Sends the number of registers to be transferred to the Series 80 computer.

Sets up a counter register for the number of times to execute label 00.

Gets a register of data to be transferred.

PROGRAM LISTING

COMMENTS

20	OUTA	Sends data.
21	ISG 00	Is there more data to transfer?
22	GTO 00	If yes, go to label 00.
23	"DATA SENT"	If no, transfer is done.
24	AVIEW	
25	SF 29	Restores comma as separator.
26	END	

Transferring HP-41 Data From A Series 80 Computer

This program will transfer HP-41 data from a Series 80 computer mass storage device to a Series 40 advanced calculator.

Instructions

1. Enter the following program (41DF85) into the HP-41. The minimum SIZE required is 001, while the actual SIZE will be determined by the amount of data being transferred.

If the data file to be transferred to the HP-41 was created by a Series 80 computer, note that it must be in the form indicated in the last paragraph of "Transferring Data in HP-41 Storage Registers to a Series 80 Computer."

2. Press key #8 to initiate subroutine 41DF85 on the Series 80 computer. The computer will ask what file name the data is stored under. Type in

41DF85

PROGRAM LISTING

```
01 LBL "41DF85"
02 CF 29
03 CF 21
04 IND
05 1 E3
06 /
07 1
08 +
09 STO 00
10 "RECEIVING"
11 AVIEW
12 LBL 00
13 IND
14 STO IND 00
15 ISG 00
16 GTO 00
17 "DONE"
18 AVIEW
19 SF 29
20 END
```

the file name and then [END LINE]. "SENDING DATA" will appear in the display.

3. Run the HP-41 program by pressing the following keys:

```
[XEQ][ALPHA]41DF85[ALPHA]
```

"RECEIVING" will appear in the display.

4. When the transfer is complete, the HP-41 display will show "DONE" and the Series 80 screen will show "DATA SENT". Each piece of data is now stored in a separate register beginning with register 01.

NOTE:

The maximum number of registers that can be transferred is 306. The time required to transfer 306 registers of data is approximately two minutes.



COMMENTS

Removes comma as separator.
Clears print enable flag.
Receives number of registers to be transferred to the HP-41.

Sets up a counter register for the number of times to execute label 00.

Receives a register of data.
Stores the data.
Is there more data to be received?
If yes, go to LBL 00.
If no, transfer is done.

Restores comma as separator.

Transfer ASCII File From HP-41 To A Series 80 Computer

Additional Equipment Required

HP 82180A Extended Functions/Memory Module
HP 82181A Extended Memory Modules (optional)

Instructions

1. The ASCII file must be in Extended Memory in the HP-41.
2. Enter the following program (41A285) into the HP-41.
3. Press key #2 on the Series 80 computer to initiate the 41A285 subroutine. "WAITING FOR DATA" will appear in the display.
4. Run the HP-41 program by pressing the following keys:
[XEQ][ALPHA]41A285[ALPHA]

41A285

PROGRAM LISTING

```
01 LBL "41A285"  
02 CF 21  
03 "FILE NAME?"  
04 AON  
05 PROMPT  
06 AOFF  
07 0  
08 SEEKPTA  
09 SF 25  
10 "SENDING DATA"  
11 AVIEW  
12 LBL 01  
13 GETREC  
14 FC? 17  
15 GTO 02  
16 LBL 03  
17 OUTA  
18 GTO 01  
19 LBL 02  
20 FS? 25  
21 GTO 03  
  
22 CLA  
23 254  
24 XTOA  
25 OUTA  
26 "DATA SENT"  
27 AVIEW  
28 END
```

The HP-41 will ask for the name of the ASCII file to be transferred. Type in the file name and press [R/S]. The display will show "SENDING DATA" and then "DATA SENT" when the transfer is complete.

5. When the Series 80 computer has received the entire file, it will beep and ask for the name under which you want the file stored. Type in the file name and press [END LINE]. "ASCII FILE STORED" will be displayed when this is completed.

NOTE:

It takes approximately five minutes to transfer and store a 600 register ASCII file. ASCII character 254 is sent by the HP-41 to signal the end of the file.

COMMENTS

Clears print enable flag.

Positions the pointers at the beginning of the file.
Sets error ignore flag.

Gets data.
Is this the end of record?
If yes, go to label 02.
If no, send contents of the alpha register.

Get another record.

Is this the end of file?
If no, go to label 3 and send contents of the alpha register.
If yes, send character 254 to signal the end of file.

Transferring An ASCII File From A Series 80 Computer To An HP-41

Additional Equipment Required

HP 82180A Extended Functions/Memory Module

HP 82181A Extended Memory Modules (optional)

Instructions

1. The ASCII file must be stored on a Series 80 mass storage medium. If the data file to be transferred to the HP-41 was created by a Series 80 computer, the end of each record must be indicated by a line feed, with character 254 used to signal the end of the file.
2. Enter the following program (41AF85) into the HP-41. This program requires a minimum SIZE of 002.
3. Press key #9 on the Series 80 computer to initiate the subroutine 41AF85. The computer will ask you for the name of the file under which the ASCII information is stored. Type in the file name and press [END LINE]. "SENDING DATA" will be displayed on the screen.

41AF85

PROGRAM LISTING

```
01 LBL "41AF85"
02 CF 21
03 CF 26
04 "FILE NAME?"
05 AON
06 PROMPT
07 IND

08 AOFF
09 CRFLAS
10 "RECEIVING"
11 AVIEW
12 LBL 05
13 IND

14 X=0?
15 GTO 03
16 INA
```

4. Run the HP-41 program by typing the following:

```
[XEQ][ALPHA]41AF85[ALPHA]
```

The HP-41 will ask for the name the ASCII file is to be stored under. Type in the name of the file and press [R/S].

The HP-41 will display "RECEIVING" and the Series 80 computer will show the message "SENDING DATA".

5. When the transfer is complete the HP-41 will display "DONE" and the Series 80 computer will display "ASCII FILE SENT".

NOTE:

It takes approximately five minutes to transfer and store a 600 register ASCII file. ASCII character 254 signals the end of the file.

COMMENTS

Clears print enable flag.
Disables beeper.

Receives number of registers necessary to create the ASCII file.

Creates the ASCII file.

Receives the number of times the contents of the alpha register is to be sent for that record.

Are there more alpha register contents to be sent?

If yes, go to label 03; transaction completed.

If no, inputs the ASCII data into the alpha register.

PROGRAM LISTING

COMMENTS

17	1	
18	-	Are the contents of the alpha register
19	X=0?	for the record to be sent once?
20	GTO 02	If yes, go to label 02 and append the
		record.
21	1 E3	If no, contents of the alpha register
		are to be sent more than once.
22	/	
23	1	Sets up appropriate number of times
24	+	to append characters to the record.
25	STO 00	
26	APPREC	Store ASCII data in Extended Memory.
27	LBL 01	
28	INA	
29	APPCHR	Store ASCII data in current record
		in Extended Memory.
30	ISG 00	Is there more ASCII data for this
		record?
31	GTO 01	If yes, get ASCII data for this
		record.
32	GTO 05	If no, get ASCII data for the next
		record.
33	LBL 02	
34	APPREC	Store ASCII data in Extended Memory.
35	GTO 05	Go get more ASCII data.
36	LBL 03	
37	"DONE"	
38	AVIEW	
39	END	

Transferring An HP-41 Program To A Series 80 Computer

Additional Equipment Required

HP 82183A Extended I/O Module

Instructions

1. The program to be transferred must be in the HP-41 main memory and have a global label.
2. Enter the following program (41P285) into the HP-41.
3. Press key #3 on the Series 80 computer to initiate the 41P285 sub-routine. "WAITING FOR PROGRAM" will be displayed.
4. Run the HP-41 program by pressing the following keys:

[XEQ][ALPHA]41P285[ALPHA]

The HP-41 will ask for the name (i.e. the global label) of the program to be transferred. Key in the name and press [R/S]. The following messages will appear in the display:

"SENDING PROG." followed by "PROG. SENT" when the transfer is complete.

5. The Series 80 computer will beep and ask you for the name under which you want to store the program. Type in the name and press [END LINE]. "PROGRAM STORED" will appear in the display when the transaction is complete. The program will be stored in a data file.

NOTE:

It takes approximately one minute to transfer a program of about 2,100 bytes.

41P285

	PROGRAM LISTING	COMMENTS
01	LBL "41P285"	
02	CF 21	
03	SF 17	Clears print enable flag. Suppresses carriage return/line feed with OUTA.
04	"PROG. NAME?"	
05	AON	
06	PROMPT	
07	AOFF	
08	ASTO X	Store program name in X.
09	"SENDING PROG"	
10	AVIEW	
11	CLA	
12	ARCL X	Recall program name to the alpha register.
13	OUTP	Sends program data byte values.
14	CLA	
15	10	
16	XTOAL	Terminates Series 80 ENTER statement with a line feed.
17	OUTA	
18	"PROG. SENT"	
19	AVIEW	
20	END	

Transferring An HP-41 Program From A Series 80 Computer

Additional Equipment Required

HP 82183A Extended I/O Module

Instructions

1. The HP-41 program must be stored on a Series 80 computer mass storage medium.
2. Enter the following program (41PF85) into the HP-41 and press the following keys ([g] indicates the gold shift key):

[g][GTO][.][.]
3. Press key #10 on the Series 80 computer to initiate the 41PF85 subroutine. The computer will ask for the name of the file under which the program to be transferred is

41PF85

PROGRAM LISTING

```
01 LBL "41PF85"  
02 CF 21  
03 "RECEIVING"  
04 AVIEW  
05 INP  
06 "DONE"  
07 AVIEW  
08 END
```

stored (see note, below). Type in the file name and press [END LINE]. "SENDING PROGRAM" will appear on the screen.

4. Run the HP-41 program by pressing the following keys:

[XEQ][ALPHA]41PF85[ALPHA]

"RECEIVING" will appear in the display.
5. When the transfer is complete, "PROGRAM SENT" will appear on the Series 80 display and "DONE" will show on the HP-41.

NOTE:

The program which was transferred to the Series 80 mass storage device retains its original name.

It takes about two minutes to transfer a program of approximately 2,100 bytes.

COMMENTS

Clears print enable flag.

Receives program data byte values.

Printing HP-41 Programs On A Series 80 Printer

Instructions

1. Press key #4 on the Series 80 computer to initiate the 41PRP subroutine. The message "WAITING" will appear on the screen.

2. Press the following keys to print a program:

[XEQ][ALPHA]MANIO[ALPHA]

1[XEQ][ALPHA]SELECT[ALPHA]

[XEQ][ALPHA]PRP[ALPHA][ALPHA](name of program to be printed)[ALPHA]

3. When the printing has been completed, "DONE" will appear on the Series 80 display.

NOTE: The HP-41 program must contain an END statement.

It takes approximately 12 minutes to print a 2,233 byte program on an HP 82905B printer.

